
Parser

• performs context-free syntax analysis
• guides context-sensitive analysis
• constructs an intermediate representation
• produces meaningful error messages
• attempts error correction

Parser construction, a brief overview (in two classes).

Syntax analysis

Context-free syntax is specified with a context-free grammar.

Formally, a CFG G is a 4-tuple (Vt,Vn,S,P), where:

Vt is the set of terminal symbols in the grammar.
For our purposes, Vt is the set of tokens returned by the scanner.

Vn, the nonterminals, is a set of syntactic variables that denote sets of
(sub)strings occurring in the language.
These are used to impose a structure on the grammar.

S is a distinguished nonterminal (S ∈Vn) denoting the entire set of strings
in L(G).
This is sometimes called a goal/start symbol.

P is a finite set of productions specifying how terminals and non-terminals
can be combined to form strings in the language.
Each production must have a single non-terminal on its left hand side.

The set V = Vt ∪Vn is called the vocabulary of G

Notation and terminology

• a,b,c, . . . ∈Vt

• A,B,C, . . . ∈Vn

• U,V,W, . . . ∈V

• α,β,γ, . . . ∈V ∗

• u,v,w, . . . ∈V ∗t

If A→ γ then αAβ⇒ αγβ is a single-step derivation using A→ γ

Similarly,⇒∗ and⇒+ denote derivations of ≥ 0 and ≥ 1 steps

If S⇒∗ β then β is said to be a sentential form of G

L(G) = {w ∈V ∗t | S⇒
+ w}, w ∈ L(G) is called a sentence of G

Note, L(G) = {β ∈V ∗ | S⇒∗ β}∩V∗t

Syntax analysis

Grammars are often written in Backus-Naur form (BNF).

Example:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈expr〉〈op〉〈expr〉
3 | num

4 | id

5 〈op〉 ::= +
6 | −
7 | ∗
8 | /

This describes simple expressions over numbers and identifiers.

In a BNF for a grammar, we represent

1. non-terminals with angle brackets or capital letters
2. terminals with typewriter font or underline
3. productions as in the example

Scanning vs. parsing

Where do we draw the line?

term ::= [a−zA−z]([a−zA−z] | [0−9])∗

| 0 | [1−9][0−9]∗

op ::= + | − | ∗ | /
expr ::= (term op)∗term

Regular expressions are used to classify:

• identifiers, numbers, keywords

• REs are more concise and simpler for tokens than a grammar

• more efficient scanners can be built from REs (DFAs) than grammars

Context-free grammars are used to count (anbn):

• brackets: (), begin. . . end, if. . . then. . . else
• imparting structure: expressions

Syntactic analysis is complicated enough: grammar for C has around 200
productions. Factoring out lexical analysis as a separate phase makes
compiler more manageable.

Derivations

We can view the productions of a CFG as rewriting rules.

Using our example CFG:

〈goal〉 ⇒ 〈expr〉
⇒ 〈expr〉〈op〉〈expr〉
⇒ 〈expr〉〈op〉〈expr〉〈op〉〈expr〉
⇒ 〈id,x〉〈op〉〈expr〉〈op〉〈expr〉
⇒ 〈id,x〉+ 〈expr〉〈op〉〈expr〉
⇒ 〈id,x〉+ 〈num,2〉〈op〉〈expr〉
⇒ 〈id,x〉+ 〈num,2〉∗ 〈expr〉
⇒ 〈id,x〉+ 〈num,2〉∗ 〈id,y〉

We have derived the sentence x + 2 ∗ y.
We denote this 〈goal〉⇒∗ id + num ∗ id.

Such a sequence of rewrites is a derivation or a parse.

The process of discovering a derivation is called parsing.

Derivations

At each step, we chose a non-terminal to replace.

This choice can lead to different derivations.

Two are of particular interest:

leftmost derivation
the leftmost non-terminal is replaced at each step

rightmost derivation
the rightmost non-terminal is replaced at each step

The previous example was a leftmost derivation.

Rightmost derivation

For the string x + 2 ∗ y:

〈goal〉 ⇒ 〈expr〉
⇒ 〈expr〉〈op〉〈expr〉
⇒ 〈expr〉〈op〉〈id,y〉
⇒ 〈expr〉∗ 〈id,y〉
⇒ 〈expr〉〈op〉〈expr〉∗ 〈id,y〉
⇒ 〈expr〉〈op〉〈num,2〉∗ 〈id,y〉
⇒ 〈expr〉+ 〈num,2〉∗ 〈id,y〉
⇒ 〈id,x〉+ 〈num,2〉∗ 〈id,y〉

Again, 〈goal〉⇒∗ id + num ∗ id.

Precedence

goal

expr

expr op expr

expr op expr * <id,y>

<num,2>+<id,x>

Treewalk evaluation computes (x + 2) ∗ y
— the “wrong” answer!

Should be x + (2 ∗ y)

Precedence

These two derivations point out a problem with the grammar.

It has no notion of precedence, or implied order of evaluation.

To add precedence takes additional machinery:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈expr〉+ 〈term〉
3 | 〈expr〉−〈term〉
4 | 〈term〉
5 〈term〉 ::= 〈term〉 ∗ 〈factor〉
6 | 〈term〉/〈factor〉
7 | 〈factor〉
8 〈factor〉 ::= num

9 | id

This grammar enforces a precedence on the derivation:

• terms must be derived from expressions
• forces the “correct” tree

Precedence

Now, for the string x + 2 ∗ y:

〈goal〉 ⇒ 〈expr〉
⇒ 〈expr〉+ 〈term〉
⇒ 〈expr〉+ 〈term〉 ∗ 〈factor〉
⇒ 〈expr〉+ 〈term〉 ∗ 〈id,y〉
⇒ 〈expr〉+ 〈factor〉 ∗ 〈id,y〉
⇒ 〈expr〉+ 〈num,2〉∗ 〈id,y〉
⇒ 〈term〉+ 〈num,2〉∗ 〈id,y〉
⇒ 〈factor〉+ 〈num,2〉∗ 〈id,y〉
⇒ 〈id,x〉+ 〈num,2〉∗ 〈id,y〉

Again, 〈goal〉⇒∗ id + num ∗ id, but this time, we build the desired tree.

Precedence

expr

expr

+

term

factor

<id,x>

goal

term

*term

<num,2>

factor

factor

<id,y>

Treewalk evaluation computes x + (2 ∗ y)

Ambiguity

If a grammar has more than one derivation for a single sentential form,
then it is ambiguous

Example:
〈stmt〉 ::= if 〈expr〉then 〈stmt〉

| if 〈expr〉then 〈stmt〉else 〈stmt〉
| other stmts

Consider deriving the sentential form:

if E1 then if E2 then S1 else S2

It has two derivations.

This ambiguity is purely grammatical.

It is a context-free ambiguity.

Ambiguity

May be able to eliminate ambiguities by rearranging the grammar:
〈stmt〉 ::= 〈matched〉

| 〈unmatched〉
〈matched〉 ::= if 〈expr〉 then 〈matched〉 else 〈matched〉

| other stmts

〈unmatched〉 ::= if 〈expr〉 then 〈stmt〉
| if 〈expr〉 then 〈matched〉 else 〈unmatched〉

This generates the same language as the ambiguous grammar, but
applies the common sense rule:

match each else with the closest unmatched then

This is most likely the language designer’s intent.

Ambiguity

Ambiguity is often due to confusion in the context-free specification.

Context-sensitive confusions can arise from overloading.

Example:

a = f(17)

In many Algol-like languages, f could be a function or subscripted
variable.

Disambiguating this statement requires context:

• need values of declarations
• not context-free
• really an issue of type

Rather than complicate parsing, we will handle this separately.

Parsing: the big picture

parser

generator

code

parser

tokens

IR

grammar

Our goal is a flexible parser generator system

Top-down versus bottom-up

Top-down parsers

• start at the root of derivation tree and fill in

• picks a production and tries to match the input

• may require backtracking

• some grammars are backtrack-free (predictive)

Bottom-up parsers

• start at the leaves and fill in

• start in a state valid for legal first tokens

• as input is consumed, change state to encode possibilities
(recognize valid prefixes)

• use a stack to store both state and sentential forms

Top-down parsing

A top-down parser starts with the root of the parse tree, labelled with the
start or goal symbol of the grammar.

To build a parse, it repeats the following steps until the fringe of the parse
tree matches the input string

1. At a node labelled A, select a production A→ α and construct the
appropriate child for each symbol of α

2. When a terminal is added to the fringe that doesn’t match the input
string, backtrack

3. Find the next node to be expanded (must have a label in Vn)

The key is selecting the right production in step 1

⇒ should be guided by input string

Simple expression grammar

Recall our grammar for simple expressions:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈expr〉 + 〈term〉
3 | 〈expr〉 −〈term〉
4 | 〈term〉
5 〈term〉 ::= 〈term〉 ∗ 〈factor〉
6 | 〈term〉 /〈factor〉
7 | 〈factor〉
8 〈factor〉 ::= num

9 | id

Consider the input string x − 2 ∗ y

Example
Prod’n Sentential form Input

– 〈goal〉 ↑x − 2 ∗ y

1 〈expr〉 ↑x − 2 ∗ y

2 〈expr〉 + 〈term〉 ↑x − 2 ∗ y
4 〈term〉 + 〈term〉 ↑x − 2 ∗ y

7 〈factor〉 + 〈term〉 ↑x − 2 ∗ y
9 id + 〈term〉 ↑x − 2 ∗ y

– id + 〈term〉 x ↑ − 2 ∗ y

– 〈expr〉 ↑x − 2 ∗ y

3 〈expr〉 − 〈term〉 ↑x − 2 ∗ y

4 〈term〉 − 〈term〉 ↑x − 2 ∗ y
7 〈factor〉 − 〈term〉 ↑x − 2 ∗ y

9 id − 〈term〉 ↑x − 2 ∗ y
– id − 〈term〉 x ↑ − 2 ∗ y

– id − 〈term〉 x − ↑2 ∗ y
7 id − 〈factor〉 x − ↑2 ∗ y

8 id − num x − ↑2 ∗ y

– id − num x − 2 ↑ ∗ y

– id − 〈term〉 x − ↑2 ∗ y

5 id − 〈term〉 ∗ 〈factor〉 x − ↑2 ∗ y
7 id − 〈factor〉 ∗ 〈factor〉 x − ↑2 ∗ y

8 id − num ∗ 〈factor〉 x − ↑2 ∗ y
– id − num ∗ 〈factor〉 x − 2 ↑ ∗ y

– id − num ∗ 〈factor〉 x − 2 ∗ ↑y
9 id − num ∗ id x − 2 ∗ ↑y
– id − num ∗ id x − 2 ∗ y ↑

Example

Another possible parse for x − 2 ∗ y

Prod’n Sentential form Input
– 〈goal〉 ↑x − 2 ∗ y

1 〈expr〉 ↑x − 2 ∗ y

2 〈expr〉 + 〈term〉 ↑x − 2 ∗ y

2 〈expr〉 + 〈term〉 + 〈term〉 ↑x − 2 ∗ y

2 〈expr〉 + 〈term〉 + · · · ↑x − 2 ∗ y

2 〈expr〉 + 〈term〉 + · · · ↑x − 2 ∗ y

2 · · · ↑x − 2 ∗ y

If the parser makes the wrong choices, expansion doesn’t terminate.
This isn’t a good property for a parser to have.

(Parsers should terminate!)

Left-recursion

Top-down parsers cannot handle left-recursion in a grammar

Formally, a grammar is left-recursive if

∃A ∈Vn such that A⇒+ Aα for some string α

Our simple expression grammar is left-recursive

Eliminating left-recursion

To remove left-recursion, we can transform the grammar

Consider the grammar fragment:

〈foo〉 ::= 〈foo〉α
| β

where α and β do not start with 〈foo〉

We can rewrite this as:

〈foo〉 ::= β〈bar〉
〈bar〉 ::= α〈bar〉

| ε

where 〈bar〉 is a new non-terminal

This fragment contains no left-recursion

Example
Our expression grammar contains two cases of left-recursion

〈expr〉 ::= 〈expr〉+ 〈term〉
| 〈expr〉−〈term〉
| 〈term〉

〈term〉 ::= 〈term〉 ∗ 〈factor〉
| 〈term〉/〈factor〉
| 〈factor〉

Applying the transformation gives

〈expr〉 ::= 〈term〉〈expr′〉
〈expr′〉 ::= +〈term〉〈expr′〉

| ε
| −〈term〉〈expr′〉

〈term〉 ::= 〈factor〉〈term′〉
〈term′〉 ::= ∗〈factor〉〈term′〉

| ε
| /〈factor〉〈term′〉

With this grammar, a top-down parser will

• terminate
• backtrack on some inputs

Example

This cleaner grammar defines the same language

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉 + 〈expr〉
3 | 〈term〉−〈expr〉
4 | 〈term〉
5 〈term〉 ::= 〈factor〉 ∗ 〈term〉
6 | 〈factor〉/〈term〉
7 | 〈factor〉
8 〈factor〉 ::= num

9 | id

It is

• right-recursive
• free of ε-productions

Unfortunately, it generates different associativity
Same syntax, different meaning

Example

Our long-suffering expression grammar:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉〈expr′〉
3 〈expr′〉 ::= +〈term〉〈expr′〉
4 | −〈term〉〈expr′〉
5 | ε
6 〈term〉 ::= 〈factor〉〈term′〉
7 〈term′〉 ::= ∗〈factor〉〈term′〉
8 | /〈factor〉〈term′〉
9 | ε

10 〈factor〉 ::= num

11 | id

Recall, we factored out left-recursion
26



How much lookahead is needed?

We saw that top-down parsers may need to backtrack when they select
the wrong production

Do we need arbitrary lookahead to parse CFGs?

• in general, yes

• use the Earley or Cocke-Younger, Kasami algorithms

Fortunately

• large subclasses of CFGs can be parsed with limited lookahead

• most programming language constructs can be expressed in a
grammar that falls in these subclasses

Among the interesting subclasses are:

LL(1): l eft to right scan, left-most derivation, 1-token lookahead; and
LR(1): l eft to right scan, right-most derivation, 1-token lookahead

Predictive parsing

Basic idea:

For any two productions A→ α | β, we would like a distinct way of
choosing the correct production to expand.

For some RHS α ∈G, define FIRST(α) as the set of tokens that appear
first in some string derived from α.
That is, for some w ∈V ∗t , w ∈ FIRST(α) iff. α⇒∗ wγ.

Key property:
Whenever two productions A→ α and A→ β both appear in the grammar,
we would like

FIRST(α)∩ FIRST(β) = φ

This would allow the parser to make a correct choice with a lookahead of
only one symbol!

The example grammar has this property!

Left factoring

What if a grammar does not have this property?

Sometimes, we can transform a grammar to have this property.

For each non-terminal A find the longest prefix
α common to two or more of its alternatives.

if α 6= ε then replace all of the A productions
A→ αβ1 | αβ2 | · · · | αβn

with
A→ αA′

A′→ β1 | β2 | · · · | βn

where A′ is a new non-terminal.

Repeat until no two alternatives for a single
non-terminal have a common prefix.

Example

Consider a right-recursive version of the expression grammar:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉+ 〈expr〉
3 | 〈term〉−〈expr〉
4 | 〈term〉
5 〈term〉 ::= 〈factor〉 ∗ 〈term〉
6 | 〈factor〉/〈term〉
7 | 〈factor〉
8 〈factor〉 ::= num

9 | id

To choose between productions 2, 3, & 4, the parser must see past the
num or id and look at the +, −, ∗, or /.

FIRST(2)∩ FIRST(3)∩ FIRST(4) 6= φ

This grammar fails the test.

Note: This grammar is right-associative.

Example

There are two nonterminals that must be left-factored:

〈expr〉 ::= 〈term〉+ 〈expr〉
| 〈term〉−〈expr〉
| 〈term〉

〈term〉 ::= 〈factor〉 ∗ 〈term〉
| 〈factor〉/〈term〉
| 〈factor〉

Applying the transformation gives us:

〈expr〉 ::= 〈term〉〈expr′〉
〈expr′〉 ::= +〈expr〉

| −〈expr〉
| ε

〈term〉 ::= 〈factor〉〈term′〉
〈term′〉 ::= ∗〈term〉

| /〈term〉
| ε

Example

Substituting back into the grammar yields

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉〈expr′〉
3 〈expr′〉 ::= +〈expr〉
4 | −〈expr〉
5 | ε
6 〈term〉 ::= 〈factor〉〈term′〉
7 〈term′〉 ::= ∗〈term〉
8 | /〈term〉
9 | ε

10 〈factor〉 ::= num

11 | id

Now, selection requires only a single token lookahead.

Note: This grammar is still right-associative.

Example

Sentential form Input
– 〈goal〉 ↑x − 2 ∗ y
1 〈expr〉 ↑x − 2 ∗ y

2 〈term〉〈expr′〉 ↑x − 2 ∗ y
6 〈factor〉〈term′〉〈expr′〉 ↑x − 2 ∗ y

11 id〈term′〉〈expr′〉 ↑x − 2 ∗ y

– id〈term′〉〈expr′〉 x ↑- 2 ∗ y

9 idε 〈expr′〉 x ↑- 2

4 id− 〈expr〉 x ↑- 2 ∗ y
– id− 〈expr〉 x − ↑2 ∗ y

2 id− 〈term〉〈expr′〉 x − ↑2 ∗ y
6 id− 〈factor〉〈term′〉〈expr′〉 x − ↑2 ∗ y

10 id− num〈term′〉〈expr′〉 x − ↑2 ∗ y

– id− num〈term′〉〈expr′〉 x − 2 ↑* y

7 id− num∗ 〈term〉〈expr′〉 x − 2 ↑* y

– id− num∗ 〈term〉〈expr′〉 x − 2 ∗ ↑y
6 id− num∗ 〈factor〉〈term′〉〈expr′〉 x − 2 ∗ ↑y

11 id− num∗ id〈term′〉〈expr′〉 x − 2 ∗ ↑y
– id− num∗ id〈term′〉〈expr′〉 x − 2 ∗ y↑
9 id− num∗ id〈expr′〉 x − 2 ∗ y↑
5 id− num∗ id x − 2 ∗ y↑

The next symbol determined each choice correctly.

Back to left-recursion elimination

Given a left-factored CFG, to eliminate left-recursion:

if ∃ A→ Aα then replace all of the A productions
A→ Aα | β | . . . | γ

with
A→ NA′

N→ β | . . . | γ
A′→ αA′ | ε

where N and A′ are new productions.

Repeat until there are no left-recursive productions.

Generality

Question:

By left factoring and eliminating left-recursion, can we transform
an arbitrary context-free grammar to a form where it can be
predictively parsed with a single token lookahead?

Answer:

Given a context-free grammar that doesn’t meet our conditions, it
is undecidable whether an equivalent grammar exists that does
meet our conditions.

Many context-free languages do not have such a grammar:

{an0bn | n≥ 1}
[

{an1b2n | n≥ 1}

Must look past an arbitrary number of a’s to discover the 0 or the 1 and so
determine the derivation.

Recursive descent parsing

Now, we can produce a simple recursive descent parser from the
(right-associative) grammar.

goal:

token ← next token();

if (expr() = ERROR | token 6= EOF) then

return ERROR;

expr:

if (term() = ERROR) then

return ERROR;

else return expr prime();

expr prime:

if (token = PLUS) then

token ← next token();

return expr();

else if (token = MINUS) then

token ← next token();

return expr();

else return OK;

Recursive descent parsing

term:

if (factor() = ERROR) then

return ERROR;

else return term prime();

term prime:

if (token = MULT) then

token ← next token();

return term();

else if (token = DIV) then

token ← next token();

return term();

else return OK;

factor:

if (token = NUM) then

token ← next token();

return OK;

else if (token = ID) then

token ← next token();

return OK;

else return ERROR;

Building the tree

One of the key jobs of the parser is to build an intermediate
representation of the source code.

To build an abstract syntax tree, we can simply insert code at the
appropriate points:

• factor() can stack nodes id, num

• term prime() can stack nodes ∗, /

• term() can pop 3, build and push subtree

• expr prime() can stack nodes +, −

• expr() can pop 3, build and push subtree

• goal() can pop and return tree

Non-recursive predictive parsing

Observation:

Our recursive descent parser encodes state information in its
run-time stack, or call stack.

Using recursive procedure calls to implement a stack abstraction may not
be particularly efficient.

This suggests other implementation methods:

• explicit stack, hand-coded parser

• stack-based, table-driven parser

Non-recursive predictive parsing

Now, a predictive parser looks like:

scanner
table-driven

parser
IR

parsing

tables

stack

source

code

tokens

Rather than writing code, we build tables.

Building tables can be automated!

Table-driven parsers

A parser generator system often looks like:

scanner
table-driven

parser
IR

parsing

tables

stack

source

code

tokens

parser

generator
grammar

This is true for both top-down (LL) and bottom-up (LR) parsers

Non-recursive predictive parsing

Input: a string w and a parsing table M for G

tos ← 0

Stack[tos] ← EOF

Stack[++tos] ← Start Symbol
token ← next token()

repeat

X ← Stack[tos]

if X is a terminal or EOF then

if X = token then

pop X

token ← next token()

else error()

else /* X is a non-terminal */

if M[X,token] = X → Y1Y2· · ·Yk then

pop X

push Yk,Yk−1, · · · ,Y1
else error()

until X = EOF

Non-recursive predictive parsing

What we need now is a parsing table M.

Our expression grammar:

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉〈expr′〉
3 〈expr′〉 ::= +〈expr〉
4 | −〈expr〉
5 | ε
6 〈term〉 ::= 〈factor〉〈term′〉
7 〈term′〉 ::= ∗〈term〉
8 | /〈term〉
9 | ε

10 〈factor〉 ::= num

11 | id

Its parse table:

id num + − ∗ / $†

〈goal〉 1 1 – – – – –
〈expr〉 2 2 – – – – –
〈expr′〉 – – 3 4 – – 5
〈term〉 6 6 – – – – –
〈term′〉 – – 9 9 7 8 9
〈factor〉 11 10 – – – – –

† we use $ to represent EOF

FIRST

For a string of grammar symbols α, define FIRST(α) as:

• the set of terminal symbols that begin strings derived from α:
{a ∈Vt | α⇒∗ aβ}

• If α⇒∗ ε then ε ∈ FIRST(α)

FIRST(α) contains the set of tokens valid in the initial position in α

To build FIRST(X):

1. If X ∈Vt then FIRST(X) is {X}

2. If X → ε then add ε to FIRST(X)

3. If X → Y1Y2· · ·Yk:

(a) Put FIRST(Y1)−{ε} in FIRST(X)

(b) ∀i : 1 < i≤ k, if ε ∈ FIRST(Y1)∩·· ·∩ FIRST(Yi−1)

(i.e., Y1 · · ·Yi−1⇒
∗ ε)

then put FIRST(Yi)−{ε} in FIRST(X)

(c) If ε ∈ FIRST(Y1)∩·· ·∩ FIRST(Yk) then put ε in FIRST(X)

Repeat until no more additions can be made.

FOLLOW

For a non-terminal A, define FOLLOW(A) as

the set of terminals that can appear immediately to the right of A

in some sentential form

Thus, a non-terminal’s FOLLOW set specifies the tokens that can legally
appear after it.

A terminal symbol has no FOLLOW set.

To build FOLLOW(A):

1. Put $ in FOLLOW(〈goal〉)

2. If A→ αBβ:

(a) Put FIRST(β)−{ε} in FOLLOW(B)

(b) If β = ε (i.e., A→ αB) or ε ∈ FIRST(β) (i.e., β⇒∗ ε) then put
FOLLOW(A) in FOLLOW(B)

Repeat until no more additions can be made

LL(1) grammars

Previous definition

A grammar G is LL(1) iff. for all non-terminals A, each distinct pair
of productions A→ β and A→ γ satisfy the condition
FIRST(β)

T

FIRST(γ) = φ.

What if A⇒∗ ε?

Revised definition

A grammar G is LL(1) iff. for each set of productions
A→ α1 | α2 | · · · | αn:

1. FIRST(α1),FIRST(α2), . . . ,FIRST(αn) are all pairwise disjoint

2. If αi⇒
∗ ε then FIRST(α j)

T

FOLLOW(A) = φ,∀1≤ j ≤ n, i 6= j.

If G is ε-free, condition 1 is sufficient.

LL(1) grammars

Provable facts about LL(1) grammars:

1. No left-recursive grammar is LL(1)

2. No ambiguous grammar is LL(1)

3. Some languages have no LL(1) grammar

4. A ε–free grammar where each alternative expansion for A begins with
a distinct terminal is a simple LL(1) grammar.

Example

• S→ aS | a is not LL(1) because FIRST(aS) = FIRST(a) = {a}

• S→ aS′

S′→ aS′ | ε
accepts the same language and is LL(1)

LL(1) parse table construction

Input: Grammar G

Output: Parsing table M

Method:

1. ∀ productions A→ α:

(a) ∀a ∈ FIRST(α), add A→ α to M[A,a]

(b) If ε ∈ FIRST(α):

i. ∀b ∈ FOLLOW(A), add A→ α to M[A,b]

ii. If $∈ FOLLOW(A) then add A→ α to M[A,$]

2. Set each undefined entry of M to error

If ∃M[A,a] with multiple entries then grammar is not LL(1).

Note: recall a,b ∈Vt, so a,b 6= ε

Example

Our long-suffering expression grammar:

S→ E T → FT ′

E→ T E ′ T ′→∗T | /T | ε
E ′→+E | −E | ε F → id | num

FIRST FOLLOW

S {num,id} {$}
E {num,id} {$}
E ′ {ε,+,−} {$}
T {num,id} {+,−,$}
T ′ {ε,∗,/} {+,−,$}
F {num,id} {+,−,∗,/,$}
id {id} −
num {num} −
∗ {∗} −
/ {/} −
+ {+} −
− {−} −

id num + − ∗ / $
S S→ E S→ E − − − − −
E E→ T E ′ E→ T E ′ − − − − −
E ′ − − E ′→+E E ′→−E − − E ′→ ε
T T → FT ′ T → FT ′ − − − − −
T ′ − − T ′→ ε T ′→ ε T ′→∗T T ′→ /T T ′→ ε
F F → id F → num − − − − −

Building the tree

Again, we insert code at the right points:

tos ← 0
Stack[tos] ← EOF
Stack[++tos] ← root node
Stack[++tos] ← Start Symbol
token ← next token()
repeat

X ← Stack[tos]
if X is a terminal or EOF then

if X = token then
pop X
token ← next token()
pop and fill in node

else error()
else /* X is a non-terminal */

if M[X,token] = X → Y1Y2 · · ·Yk then
pop X
pop node for X
build node for each child and
make it a child of node for X
push nk,Yk,nk−1,Yk−1, · · · ,n1,Y1

else error()
until X = EOF

A grammar that is not LL(1)

〈stmt〉 ::= if 〈expr〉 then 〈stmt〉
| if 〈expr〉 then 〈stmt〉 else 〈stmt〉
| . . .

Left-factored:

〈stmt〉 ::= if 〈expr〉 then 〈stmt〉 〈stmt′〉 | . . .
〈stmt′〉 ::= else 〈stmt〉 | ε

Now, FIRST(〈stmt′〉) = {ε,else}
Also, FOLLOW(〈stmt′〉) = {else,$}
But, FIRST(〈stmt′〉)

T

FOLLOW(〈stmt′〉) = {else} 6= φ

On seeing else, conflict between choosing

〈stmt′〉 ::= else 〈stmt〉 and 〈stmt′〉 ::= ε

⇒ grammar is not LL(1)!

The fix:

Put priority on 〈stmt′〉 ::= else 〈stmt〉 to associate else with
closest previous then.

Error recovery

Key notion:

• For each non-terminal, construct a set of terminals on which the
parser can synchronize

• When an error occurs looking for A, scan until an element of
SYNCH(A) is found

Building SYNCH:

1. a ∈ FOLLOW(A)⇒ a ∈ SYNCH(A)

2. place keywords that start statements in SYNCH(A)

3. add symbols in FIRST(A) to SYNCH(A)

If we can’t match a terminal on top of stack:

1. pop the terminal

2. print a message saying the terminal was inserted

3. continue the parse

(i.e., SYNCH(a) = Vt−{a})

Some definitions

Recall

For a grammar G, with start symbol S, any string α such that S⇒∗ α is
called a sentential form

• If α ∈V ∗t , then α is called a sentence in L(G)

• Otherwise it is just a sentential form (not a sentence in L(G))

A left-sentential form is a sentential form that occurs in the leftmost
derivation of some sentence.

A right-sentential form is a sentential form that occurs in the rightmost
derivation of some sentence.

Bottom-up parsing

Goal:

Given an input string w and a grammar G, construct a parse tree
by starting at the leaves and working to the root.

The parser repeatedly matches a right-sentential form from the language
against the tree’s upper frontier.

At each match, it applies a reduction to build on the frontier:

• each reduction matches an upper frontier of the partially built tree to
the RHS of some production

• each reduction adds a node on top of the frontier

The final result is a rightmost derivation, in reverse.

Example

Consider the grammar

1 S → aABe
2 A → Abc
3 | b

4 B → d

and the input string abbcde

Prod’n. Sentential Form
3 a b bcde

2 a Abc de

4 aA d e

1 aABe
– S

The trick appears to be scanning the input and finding valid sentential
forms.

Handles

What are we trying to find?

A substring α of the tree’s upper frontier that

matches some production A→ α where reducing α to A is one
step in the reverse of a rightmost derivation

We call such a string a handle.

Formally:

a handle of a right-sentential form γ is a production A→ β and a
position in γ where β may be found and replaced by A to produce
the previous right-sentential form in a rightmost derivation of γ.

i.e., if S⇒∗rm αAw⇒rm αβw then A→ β in the position following α is a
handle of αβw

Because γ is a right-sentential form, the substring to the right of a handle
contains only terminal symbols.

Handles

S

α

A

wβ
The handle A→ β in the parse tree for αβw

Handles

Theorem:

If G is unambiguous then every right-sentential form has a unique
handle.

Proof: (by definition)

1. G is unambiguous⇒ rightmost derivation is unique

2. ⇒ a unique production A→ β applied to take γi−1 to γi

3. ⇒ a unique position k at which A→ β is applied

4. ⇒ a unique handle A→ β

Example

The left-recursive expression grammar (original form)

1〈goal〉 ::= 〈expr〉
2〈expr〉 ::= 〈expr〉+ 〈term〉
3 | 〈expr〉−〈term〉
4 | 〈term〉
5〈term〉 ::= 〈term〉 ∗ 〈factor〉
6 | 〈term〉/〈factor〉
7 | 〈factor〉
8〈factor〉 ::= num

9 | id

Prod’n. Sentential Form
– 〈goal〉
1 〈expr〉
3 〈expr〉 − 〈term〉
5 〈expr〉 − 〈term〉 ∗ 〈factor〉
9 〈expr〉 − 〈term〉 ∗ id
7 〈expr〉 − 〈factor〉 ∗ id
8 〈expr〉 − num ∗ id
4 〈term〉 − num ∗ id

7 〈factor〉 − num ∗ id

9 id − num ∗ id

Handle-pruning

The process to construct a bottom-up parse is called handle-pruning.

To construct a rightmost derivation

S = γ0⇒ γ1⇒ γ2⇒ ··· ⇒ γn−1⇒ γn = w

we set i to n and apply the following simple algorithm

for i = n downto 1

1. find the handle Ai→ βi in γi

2. replace βi with Ai to generate γi−1

This takes 2n steps, where n is the length of the derivation

Stack implementation

One scheme to implement a handle-pruning, bottom-up parser is called a
shift-reduce parser.

Shift-reduce parsers use a stack and an input buffer

1. initialize stack with $

2. Repeat until the top of the stack is the goal symbol and the input
token is $

a) find the handle
if we don’t have a handle on top of the stack, shift an input symbol
onto the stack

b) prune the handle
if we have a handle A→ β on the stack, reduce

i) pop | β | symbols off the stack

ii) push A onto the stack

Example: back to x − 2 ∗ y

1〈goal〉 ::= 〈expr〉
2〈expr〉 ::= 〈expr〉+ 〈term〉
3 | 〈expr〉−〈term〉
4 | 〈term〉
5〈term〉 ::= 〈term〉 ∗ 〈factor〉
6 | 〈term〉/〈factor〉
7 | 〈factor〉
8〈factor〉 ::= num

9 | id

Stack Input Action
$ id − num ∗ id shift
$id − num ∗ id reduce 9
$〈factor〉 − num ∗ id reduce 7
$〈term〉 − num ∗ id reduce 4
$〈expr〉 − num ∗ id shift
$〈expr〉 − num ∗ id shift
$〈expr〉 − num ∗ id reduce 8
$〈expr〉 − 〈factor〉 ∗ id reduce 7
$〈expr〉 − 〈term〉 ∗ id shift
$〈expr〉 − 〈term〉 ∗ id shift
$〈expr〉 − 〈term〉 ∗ id reduce 9
$〈expr〉 − 〈term〉 ∗ 〈factor〉 reduce 5
$〈expr〉 − 〈term〉 reduce 3
$〈expr〉 reduce 1
$〈goal〉 accept

1. Shift until top of stack is the right end of a handle

2. Find the left end of the handle and reduce

5 shifts + 9 reduces + 1 accept

Shift-reduce parsing

Shift-reduce parsers are simple to understand

A shift-reduce parser has just four canonical actions:

1. shift — next input symbol is shifted onto the top of the stack

2. reduce — right end of handle is on top of stack;
locate left end of handle within the stack;
pop handle off stack and push appropriate non-terminal LHS

3. accept — terminate parsing and signal success

4. error — call an error recovery routine

Key insight: recognize handles with a DFA:

• DFA transitions shift states instead of symbols

• accepting states trigger reductions

63



LR parsing

The skeleton parser:

push s0
token ← next token()

repeat forever
s ← top of stack

if action[s,token] = "shift si" then
push si
token ← next token()

else if action[s,token] = "reduce A→ β"
then
pop | β | states
s′← top of stack
push goto[s′,A]

else if action[s, token] = "accept" then
return

else error()

This takes k shifts, l reduces, and 1 accept, where k is the length of the
input string and l is the length of the reverse rightmost derivation

Example tables

state ACTION GOTO
id + ∗ $ 〈expr〉 〈term〉 〈factor〉

0 s4 – – – 1 2 3
1 – – – acc – – –
2 – s5 – r3 – – –
3 – r5 s6 r5 – – –
4 – r6 r6 r6 – – –
5 s4 – – – 7 2 3
6 s4 – – – – 8 3
7 – – – r2 – – –
8 – r4 – r4 – – –

The Grammar
1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉+ 〈expr〉
3 | 〈term〉
4 〈term〉 ::= 〈factor〉 ∗ 〈term〉
5 | 〈factor〉
6 〈factor〉 ::= id

Note: This is a simple little right-recursive grammar; not the same as in previous lectures.

Shift-reduce parsing table construction
The Grammar

1 〈goal〉 ::= 〈expr〉
2 〈expr〉 ::= 〈term〉+ 〈expr〉
3 | 〈term〉
4 〈term〉 ::= 〈factor〉 ∗ 〈term〉
5 | 〈factor〉
6 〈factor〉 ::= id

I0 : 〈goal〉 ::= .〈expr〉$
〈expr〉 ::= .〈term〉+ 〈expr〉
〈expr〉 ::= .〈term〉
〈term〉 ::= .〈factor〉 ∗ 〈term〉
〈term〉 ::= .〈factor〉
〈factor〉 ::= .id

I1 : 〈goal〉 ::= 〈expr〉.$

I2 : 〈expr〉 ::= 〈term〉.+ 〈expr〉
〈expr〉 ::= 〈term〉.

I3 : 〈term〉 ::= 〈factor〉.∗ 〈term〉
〈term〉 ::= 〈factor〉.

I4 : 〈factor〉 ::= id.

I5 : 〈expr〉 ::= 〈term〉+ .〈expr〉
〈expr〉 ::= .〈term〉+ 〈expr〉
〈term〉 ::= .〈factor〉 ∗ 〈term〉
〈term〉 ::= .〈factor〉
〈factor〉 ::= .id

I6 : 〈term〉 ::= 〈factor〉∗ .〈term〉
〈term〉 ::= .〈factor〉 ∗ 〈term〉
〈term〉 ::= .〈factor〉
〈factor〉 ::= .id

I7 : 〈expr〉 ::= 〈term〉+ 〈expr〉.

I8 : 〈term〉 ::= 〈factor〉∗ 〈term〉.

Example using the tables

Stack Input Action
$ 0 id∗ id+ id$ s4
$ 0 4 ∗ id+ id$ r6
$ 0 3 ∗ id+ id$ s6
$ 0 3 6 id+ id$ s4
$ 0 3 6 4 + id$ r6
$ 0 3 6 3 + id$ r5
$ 0 3 6 8 + id$ r4
$ 0 2 + id$ s5
$ 0 2 5 id$ s4
$ 0 2 5 4 $ r6
$ 0 2 5 3 $ r5
$ 0 2 5 2 $ r3
$ 0 2 5 7 $ r2
$ 0 1 $ acc

67



LR(k) grammars

Informally, we say that a grammar G is LR(k) if, given a rightmost
derivation

S = γ0⇒ γ1⇒ γ2⇒ ··· ⇒ γn = w,

we can, for each right-sentential form in the derivation,

1. isolate the handle of each right-sentential form, and

2. determine the production by which to reduce

by scanning γi from left to right, going at most k symbols beyond the right
end of the handle of γi.

Why study LR grammars?

LR(1) grammars are often used to construct parsers.

We call these parsers LR(1) parsers.

• everyone’s favorite parser

• virtually all context-free programming language constructs can be
expressed in an LR(1) form

• LR grammars are the most general grammars parsable by a
deterministic, bottom-up parser

• efficient parsers can be implemented for LR(1) grammars

• LR parsers detect an error as soon as possible in a left-to-right scan
of the input

• LR grammars describe a proper superset of the languages
recognized by predictive (i.e., LL) parsers

LL(k): recognize use of a production A→ β seeing first k symbols of β

LR(k): recognize occurrence of β (the handle) having seen all of
what is derived from β plus k symbols of lookahead

LR parsing

Three common algorithms to build tables for an “LR” parser:

1. SLR(1)

• smallest class of grammars
• smallest tables (number of states)
• simple, fast construction

2. LR(1)

• full set of LR(1) grammars
• largest tables (number of states)
• slow, large construction

3. LALR(1)

• intermediate sized set of grammars
• same number of states as SLR(1)
• canonical construction is slow and large
• better construction techniques exist

An LR(1) parser for either Algol or Pascal has several thousand states,
while an SLR(1) or LALR(1) parser for the same language may have
several hundred states.

Left versus right recursion
Right Recursion:

• needed for termination in predictive parsers

• requires more stack space

• right associative operators

Left Recursion:

• works fine in bottom-up parsers

• limits required stack space

• left associative operators

Rule of thumb:

• right recursion for top-down parsers

• left recursion for bottom-up parsers

Key points :

• LL : Choose A→ b or A→ c ?

• LR : Choose A→ b or B→ b ?

Parsing review

Recursive descent

A hand coded recursive descent parser directly encodes a grammar
(typically an LL(1) grammar) into a series of mutually recursive
procedures. It has most of the linguistic limitations of LL(1).

LL(k)

An LL(k) parser must be able to recognize the use of a production
after seeing only the first k symbols of its right hand side.

LR(k)

An LR(k) parser must be able to recognize the occurrence of the right
hand side of a production after having seen all that is derived from
that right hand side with k symbols of lookahead.

Complexity of parsing: grammar hierarchy

LL(1)

LL(k)

LL(0)

Knuth’s algorithm: O(n)

LR(k)

LR(1)

LALR(1)

SLR(1)

LR(0)

ambiguous

α−>β
type-0:

type-1: context-sensitive

αΑβ−>αδβ
Linear-bounded automaton: PSPACE complete

type-2: context-free

Α−>α
Earley’s algorithm: O(n³)

type-3: regular

A->wX

DFA: O(n)

O(n²)

unambiguous

Note: this is a hierarchy of grammars not languages

Language vs. grammar

For example, every regular language has a grammar that is LL(1), but not
all regular grammars are LL(1). Consider:

S → ab
S → ac

Without left-factoring, this grammar is not LL(1).

